

# PRESEK

List za mlade matematike, fizike, astronome in računalnikarje

ISSN 0351-6652

Letnik 14 (1986/1987)

Številka 4

Strani 226-231

Marko Razpet:

## O UPORABI REKURZIVNIH FORMUL

Ključne besede: računalništvo, matematika.

Elektronska verzija: <http://www.presek.si/14/849-Razpet.pdf>

© 1987 Društvo matematikov, fizikov in astronomov Slovenije

© 2010 DMFA - založništvo

Vse pravice pridržane. Razmnoževanje ali reproduciranje celote ali posameznih delov brez poprejšnjega dovoljenja založnika ni dovoljeno.

## O UPORABI REKURZIVNIH FORMUL

Za začetek si pogledjmo kratek programček, ki je napisan v programskem jeziku BASIC za računalnik ZX SPECTRUM. Najbrž ga ne bo pretežko prilagoditi za drug tip računalnika.

```
10 LET m=30: LET a=33
20 PLOT 0,75: DRAW 255,0
30 LET f=PI/m
40 FOR x=0 TO 255
50     LET y=a*SIN(x*f)
60     PLOT x, y+75
70 NEXT x
```


Programček nariše na ekranu valovito pikčasto krivuljo, ki ji pravimo *sinusoida*. S programsko vrstico 20 dobimo os  $x$ , v vrstici 10 pomeni  $m$  število točk na eni polovici vala, število  $a$  pa njegovo višino. Ta dva podatka lahko sami spreminjamo, da dobimo sinusoide različnih oblik. Kdor hoče, pa naj vrine še vrstico, ki bo dodala tudi os  $y$ .

Programu pravzaprav ni kaj očitati. Če pa večkrat ponovimo njegovo izvajanje, nas kmalu začne motiti njegova počasnost. Nekako 20 sekund traja od trenutka, ko ga sprožimo, do trenutka, ko nam računalnik javi, da je delo končano. Če pa program malo bolje pogledamo, se njegovi počasnosti ni treba več čuditi. Vrstice od 40 do 70 tvorijo zanko FOR—NEXT. V njej uporabljamo funkcijo sinus (SIN), dvoje množenj in eno seštevanje, računalnik mora to opraviti kar 256-krat. Za izračun ene vrednosti funkcije sinus je treba kar nekaj dela, to pa je glavni vzrok, da se program odvija, kakor pač se. Za bralca, ki je že večč dela s prevajalniki, pa tale naloga: prevedi naš programček s prevajalnikom SOFTEK FP in izmeri hitrost izvajanja prevedenega programa.

Sicer pa dodajmo programu naslednje vrstice:

```
80 PAUSE 100: CLS
90 PLOT 0,75: DRAW 255,0
100 LET y0=0: LET y1=a*SIN f
110 LET c=2*COS f
120 FOR x=0 TO 255
130     PLOT x, y0+75
140     LET y2=c*y1-y0
150     LET y0=y1: LET y1=y2
160 NEXT x
```

Tako razširjen program še enkrat poženemo. Prvi del nariše sinusoido počasno, tako kot prej, kajti za vsako točko mora opraviti dvoje množenj in izračunati po eno vrednost funkcije sinus. Od programske vrstice 80 dalje dobimo še enkrat isto sliko, toda precej hitreje kot prvič. Vrstice od 120 do 160 tvorijo zopet zanko FOR–NEXT. V njej je le eno množenje in nekaj manj zahtevnih operacij. Vse to se seveda ponovi 256-krat. In v čem je skrivnost "uspeha", če se malo pohvalimo?




Slika 1

Čim krivulja ni sestavljena le iz daljic in krožnih lokov, moramo izračunati dovolj mnogo različnih točk in skozi nje potegnemo krivuljo ali s krivuljnikom ali po občutku.

Morda se je bralec v našem programu prvič srečal s funkcijo sinus. Nič zato! Ta funkcija spada med tako imenovane kotne ali, bolj učeno, trigonometrične funkcije. Veda, ki se ukvarja z njimi, se imenuje trigonometrija, njeni začetki pa segajo noter v staro Babilonijo in Egipt. Ni zlomek, da bi mi v dobi osvajanja vesolja ne razumeli nekaterih preprostih stvari iz poglavja o kotnih funkcijah. Treba je poudariti, da kote v matematiki in računalništvu običajno merimo v radianih. Kotna enota radian je središčni kot, ki na poljubni krožnici pripada loku, katerega dolžina je enaka dolžini polmera krožnice. Dovolj si je zapomniti, da je kot  $180^\circ$  enak kotu  $\pi$  radianov. Po navadnem sklepnem računu je potem kot  $1^\circ$  enak kotu  $\pi/180$  radianov. Kotne funkcije SIN, COS in TAN, ki jih pozna vaš SPECTRUM, zahtevajo kot v radianih. V matematiki imajo oznake  $\sin$ ,  $\cos$  in  $\text{tg}$ . V tem članku bomo imeli opravka le s prvima dvema.

Postavimo v ravnino pravokotni koordinatni sistem  $xy$ . Izberimo si enoto, nato načrtamo krožnico s središčem v izhodišču in z radijem 1. Ta krožnica bo pripravna za merjenje kotov. Središčnemu kotu  $\alpha$  pripada lok dolžine  $\alpha$ . Kote bomo merili po loku krožnice od pozitivne polovice osi  $x$ . Kote, merjene

v smeri, ki je nasprotna smeri gibanja kazalcev na uri, bomo šteli za pozitivne, če pa jih merimo v nasprotni smeri, pa negativne.



Slika 2

Prej omenjeni krožnici rečemo *kotomerna krožnica*. Kot  $a$  določa na njej točko  $T$  s koordinatama  $x$  in  $y$ . Abscisa  $x$  imenujemo kosinus kota  $a$ , ordinato  $y$  pa sinus kota  $a$ . To zapišemo takole:

$$x = \cos a, \quad y = \sin a$$

Trikotnik  $ONT$  na sliki 2 je pravokoten in po Pitagorovem izreku dobimo  $x^2 + y^2 = 1$ . V trigonometriji pišemo namesto  $(\sin a)^2$  oziroma  $(\cos a)^2$  raje  $\sin^2 a$  oziroma  $\cos^2 a$ . Tako smo dobili pravzaprav osnovno zvezo

$$(1) \quad \sin^2 a + \cos^2 a = 1$$

S slike 2 lahko razberemo tudi, da veljata naslednji formuli

$$(2) \quad \sin(-a) = -\sin a$$

$$(3) \quad \cos(-a) = \cos a$$

Funkcija  $\cos$  je torej soda funkcija, funkcija  $\sin$  pa liha.

Hitro s slike 2 odčitamo tudi nekaj preprostih vrednosti:  $\cos 0 = 1$ ,  $\sin 0 = 0$ ,  $\cos(\pi/2) = 0$ ,  $\sin(\pi/2) = 1$ , ...,  $\cos(2\pi) = 1$ ,  $\sin(2\pi) = 0$ .

Funkciji sinus in kosinus imamo na opisani način definirani za vse kote med  $-2\pi$  in  $2\pi$ . Lahko pa ju na smiselni način razširimo na vsa realna števila z naslednjima formulama

$$(4) \quad \sin(a + 2k\pi) = \sin a$$

$$(5) \quad \cos(a + 2k\pi) = \cos a$$

ki veljata za poljubno realno število  $a$  in poljubno celo število  $k$ . Zaradi formul (4) in (5) pravimo, da sta funkciji sinus in kosinus periodični z osnovno periodo  $2\pi$ .

Brez dokazov bomo navedli še formuli za sinus in kosinus vsote dveh realnih števil. To sta tako imenovana *adicijska izreka*

$$(6) \quad \sin(a + \beta) = \sin a \cos \beta + \cos a \sin \beta$$

$$(7) \quad \cos(a + \beta) = \cos a \cos \beta - \sin a \sin \beta$$

Veljata za poljubni realni števili  $a$  in  $\beta$ . Formuli (6) in (7) sta ključnega pomena za hitrejše risanje sinusoid na računalniku. Če upoštevamo še formuli (2) in (3), dobimo še

$$(8) \quad \sin(a - \beta) = \sin a \cos \beta - \cos a \sin \beta$$

$$(9) \quad \cos(a - \beta) = \cos a \cos \beta + \sin a \sin \beta$$

Tudi ti dve formuli veljata za poljubni realni števili  $a$  in  $\beta$ .

Povrnimo se na naše načrtovanje sinusoid. V najbolj splošnem primeru ima ta krivulja enačbo  $y = A \sin(\omega x + \varphi)$ . V njej nastopajo trije parametri: parameter  $A$  se imenuje *amplituda*,  $\omega$  *krožna frekvenca* in  $\varphi$  *začetna faza*. Za lepo sliko jih je treba pazljivo izbrati.

Odločimo se, da bomo funkcijske vrednosti računali v točkah  $x_0 = 0$ ,  $x_1 = \delta$ ,  $x_2 = 2\delta$ , ..., dokler bo pač krivulja še šla na papir oziroma na ekran. Število  $\delta$  je pozitivno in primerno majhno, odvisno od tega, kakšno gostoto točk bi radi imeli. Pripadajoče funkcijske vrednosti označimo  $y_k$ . Torej

$$y_k = A \sin(\omega x_k + \varphi) = A \sin(k\delta\omega + \varphi)$$

kjer je  $k = 0, 1, 2, \dots$ . Zaporedje  $y_0, y_1, y_2, \dots$  pa ima zanimivo lastnost, ki smo jo izkoristili v programu. Formuli (6) in (8) dasta tole zvezo med tremi zaporednimi členi zaporedja  $y_0, y_1, y_2, \dots$ :

$$(10) \quad y_{k+1} + y_{k-1} = 2y_k \cos(\delta\omega) \quad \text{za } k = 1, 2, 3, \dots$$

To je poseben primer tročlene rekurzivne formule. Če poznamo  $y_0$  in  $y_1$ , lahko izračunamo  $y_2$ , iz  $y_1$  in  $y_2$  dobimo  $y_3$  itd. Naj bo  $c = 2 \cos(\delta\omega)$ . Iz zveze (10) izračunamo  $y_{k+1}$  in dobimo

$$(11) \quad y_{k+1} = c y_k - y_{k-1} \quad \text{za } k = 1, 2, 3, \dots$$

Pri danih parametrih  $A$ ,  $\omega$ ,  $\varphi$  in izbranem koraku  $\delta$  je  $y_0 = A \sin(\varphi)$  in  $y_1 = A \sin(\delta\omega + \varphi)$ . Za vsak nadaljnji  $y_k$  napravimo le eno množenje in eno odštevanje.

V posebnem primeru, ko je  $\varphi = 0$ , dobimo  $y_0 = 0$  in  $y_1 = A \sin(\delta\omega)$  ter  $y_k = A \sin(k\delta\omega)$ , kar smo uporabili v programu od vrstice 100 dalje. V tem primeru dobimo sinusno krivuljo. Če pa je  $\varphi = \pi/2$ , potem imamo  $y_0 = A$  in  $y_1 = A \sin(\delta\omega + \pi/2) = A \cos(\delta\omega)$  ter  $y_k = A \sin(k\delta\omega + \pi/2) = A \cos(k\delta\omega)$ . Ustrezni program bi načrtal kosinusno krivuljo.

Bralec naj poskusi napisati program, ki bo s pomočjo formule (11) narisal hkrati sinusno in kosinusno krivuljo. Opazil bo, da je ena pravzaprav premik druge v smeri osi  $x$ . Hkrati pa naj kontrolira, če velja formula (1). Morda bo malo razočaran, a o tem malo kasneje.

Če je že treba tabelirati funkciji sinus in kosinus z danim korakom  $\delta$  istočasno, delamo raje malo drugače. Izračunati moramo, skratka, števila  $c_k = \cos(k\delta)$  in  $s_k = \sin(k\delta)$  za  $k = 0, 1, 2, \dots$  Adicijski izrek (6) nam takoj da

$$(12) \quad s_{k+1} = \sin(k\delta + \delta) = \sin(k\delta) \cos\delta + \cos(k\delta) \sin\delta$$

adicijski izrek (7) pa

$$(13) \quad c_{k+1} = \cos(k\delta + \delta) = \cos(k\delta) \cos\delta - \sin(k\delta) \sin\delta$$

Če na kratko označimo  $c = c_1 = \cos\delta$  in  $s = s_1 = \sin\delta$ , potem lahko (12) in (13) zapišemo bolj pregledno:

$$(14) \quad s_{k+1} = cs_k + sc_k$$

$$(15) \quad c_{k+1} = cc_k - ss_k \quad \text{za } k = 0, 1, 2, \dots$$

Ker poznamo  $c_0 = \cos 0 = 1$  in  $s_0 = \sin 0 = 0$ , lahko korak za korakom izračunamo  $c_1, s_1; c_2, s_2; \dots$

Za začetek poskusimo načrtati na računalniku elipso s pomočjo formul (14) in (15). V pravokotnem koordinatnem sistemu  $xy$  naj ima elipsa enačbo  $(x/a)^2 + (y/b)^2 = 1$ . Če jo primerjamo s formulo (1), se nam zdi, da bi postavili  $x/a = \cos t$  in  $y/b = \sin t$ . S tem imamo

$$(16) \quad x = a \cos t$$

$$y = b \sin t$$

V teh dveh enačbah nastopa parameter  $t$ , ki je realno število. S preprostim premislekom ugotovimo, da točka  $T$  s koordinatama  $x, y$  ravno enkrat obhodi elipso, če parameter  $t$  preteče interval od 0 do  $2\pi$ . Pravimo, da smo elipso parametrizirali s parametrom  $t$ . Z enačbama (16) je podana elipsa v parametrični obliki. Sedaj pa že lahko sestavimo program.

```

10 LET m=60: LET f=PI/m
20 LET c=COS f: LET s=SIN f
30 LET a=110: LET b=70
40 LET c0=1: LET s0=0
50 FOR t=0 TO 2*m
60   PLOT 128+a*c0,75+b*s0
70   LET c1=c*c0-s*s0
80   LET s1=c*s0+s*c0
90   LET c0=c1: LET s0=s1
100 NEXT t

```

Program se sicer ne izvede tako hitro kot ukaz CIRCLE pri vašem računalniku; upoštevati je treba, da je to le BASIC.

